Reduction of the differentiation index of a DAE

```
> restart;
  with (plots):
  with (DifferentialAlgebra);
```

$$[BelongsTo, DifferentialRing, Equations, Get, Inequations, Is, NormalForm, PowerSeriesSolution, ReducedForm, RosenfeldGroebner, Tools] \quad (1)$$

A DAE of differentiation index 2.
Borrowed from the book of E. Hairer and G. Wanner, vol. 2, chap. VII
Two ODE, one algebraic relation

```
> book_syst := [ diff (x(t),t) = a*y(t) + sin(b*z(t)),
                 diff (y(t),t) = 2*a*x(t) + cos(b*z(t)),
                 x(t)^2 + y(t)^2 = 1 ];
```

$$book\_syst := \left[\frac{d}{dt} x(t) = a\, y(t) + \sin(b\, z(t)),\ \frac{d}{dt} y(t) = 2\, a\, x(t) + \cos(b\, z(t)),\ x(t)^2 + y(t)^2 = 1 \right] \quad (2)$$

The numerical values are picked from the book

```
> num_book_syst := subs (a = 0.7, b = 2.5, book_syst);
```

$$num\_book\_syst := \left[\frac{d}{dt} x(t) = 0.7\, y(t) + \sin(2.5\, z(t)),\ \frac{d}{dt} y(t) = 1.4\, x(t) + \cos(2.5\, z(t)),\ x(t)^2 + y(t)^2 = 1 \right] \quad (3)$$

Three dependent variables

```
> vars := {x(t), y(t), z(t)};
```

$$vars := \{x(t), y(t), z(t)\} \quad (4)$$

Some initial values

```
> iv := x(0)=1, y(0)=0, z(0)=1;
```

$$iv := x(0) = 1,\ y(0) = 0,\ z(0) = 1 \quad (5)$$

The DAE solver of MAPLE complains.
There may exist hidden algebraic relations between initial values

```
> dsolve ({ op (num_book_syst), iv }, vars, numeric);
Error, (in dsolve/numeric/DAE/checkconstraints) the initial conditions do not satisfy the algebraic constraints
  error = 5.98, tolerance = .200e-4, constraint = 2*y(t)*(7*x(t)+5*cos(2.5000000000000000000*z(t)))+x(t)*(7*y(t)+10*sin(2.5000000000000000000*z(t)))
```

Differential elimination permits us to compute the **missing ODE** and a **complete set of constraints on initial values.**

The system is first converted to a polynomial form by introducing two new variables **s(t)** for the sine and **c(t)** for the cosine.

```
> syst := [ diff (x(t),t) = a*y(t) + s(t),
            diff (y(t),t) = 2*a*x(t) + c(t),
            x(t)^2 + y(t)^2 = 1,
            diff (s(t),t) = b*diff (z(t),t)*c(t),
            diff (c(t),t) = - b*diff(z(t),t)*s(t),
            s(t)^2 + c(t)^2 = 1];
```

$$syst := \left[\frac{d}{dt} x(t) = a\, y(t) + s(t),\ \frac{d}{dt} y(t) = 2\, a\, x(t) + c(t),\ x(t)^2 + y(t)^2 = 1,\ \frac{d}{dt} s(t) = b \left(\frac{d}{dt} z(t) \right) c(t),\ \frac{d}{dt} c(t) = -b \left(\frac{d}{dt} z(t) \right) s(t),\ s(t)^2 + c(t)^2 = 1 \right] \quad \textbf{(6)}$$

The context of the computation

```
> R := DifferentialRing
       (derivations = [t], blocks = [[y,x,s,c,z],[a(),b()]]);
```

$$R := differential\_ring \quad \textbf{(7)}$$

The differential elimination

```
> result := RosenfeldGroebner
       (syst, R, basefield = field (generators = [a,b]));
```

$$result := [regular\_differential\_chain] \quad \textbf{(8)}$$

The equations are hidden in a MAPLE table

```
> Equations (result);
```

$$\Big[\Big[108\, x(t)^3\, s(t)\, c(t)^2\, a^4 - 162\, x(t)^2\, s(t)\, c(t)^5\, a^3 + 81\, x(t)^2\, s(t)\, c(t)^3\, a^5 \quad \textbf{(9)}$$

$$+ 126\, x(t)^2\, s(t)\, c(t)^3\, a^3 - 27\, x(t)^2\, s(t)\, c(t)\, a^3 + 162\, x(t)\, s(t)\, c(t)^4\, a^4$$

$$+ 9\, x(t)\, s(t)\, c(t)^4\, a^2 - 243\, x(t)\, s(t)\, c(t)^2\, a^6 - 27\, x(t)\, s(t)\, c(t)^2\, a^4$$

$$- 12\, x(t)\, s(t)\, c(t)^2\, a^2 - 6\, x(t)^2\, s(t)\, c(t)\, a - 243\, x(t)^3\, s(t)\, c(t)^4\, a^4$$

$$+ 243\, x(t)^3\, s(t)\, c(t)^2\, a^6 - 9\, x(t)^3\, s(t)\, a^2 + 6\, x(t)\, s(t)\, a^2 + 7\, s(t)\, c(t)\, a$$

$$+ 9 \left(\frac{d}{dt} c(t) \right) c(t)^5\, a^2 - 9 \left(\frac{d}{dt} c(t) \right) c(t)^3\, a^2 + 108\, s(t)\, c(t)^5\, a^3$$

$$- 81\, s(t)\, c(t)^3\, a^5 - 54\, s(t)\, c(t)^3\, a^3 - 6\, s(t)\, c(t)^3\, a + 9\, s(t)\, c(t)\, a^3$$

$$- 27\, x(t)^3\, s(t)\, a^4 + 27\, x(t)\, s(t)\, a^4 - x(t)\, s(t) + \left(\frac{d}{dt} c(t) \right) c(t),\ x(t)$$

$$+ b \left(\frac{d}{dt} z(t)\right) c(t) + 9 \left(\frac{d}{dt} z(t)\right) c(t)^5 a^2 b - 9 \left(\frac{d}{dt} z(t)\right) c(t)^3 a^2 b$$
$$+ 243 x(t)^3 c(t)^4 a^4 - 243 x(t)^3 c(t)^2 a^6 - 108 x(t)^3 c(t)^2 a^4 + 162 x(t)^2 c(t)^5 a^3$$
$$- 81 x(t)^2 c(t)^3 a^5 - 126 x(t)^2 c(t)^3 a^3 + 27 x(t)^2 c(t) a^3 + 6 x(t)^2 c(t) a$$
$$- 162 x(t) c(t)^4 a^4 - 9 x(t) c(t)^4 a^2 + 243 x(t) c(t)^2 a^6 + 27 x(t) c(t)^2 a^4$$
$$+ 12 x(t) c(t)^2 a^2 + 27 x(t)^3 a^4 + 9 x(t)^3 a^2 - 27 x(t) a^4 - 6 x(t) a^2$$
$$- 108 c(t)^5 a^3 + 81 c(t)^3 a^5 + 54 c(t)^3 a^3 + 6 c(t)^3 a - 9 c(t) a^3 - 7 c(t) a,$$
$$y(t) c(t)^3 - y(t) c(t) - 9 x(t)^3 s(t) a^2 - 3 x(t)^2 s(t) c(t) a + x(t) s(t) c(t)^2$$
$$+ 9 x(t) s(t) a^2 - x(t) s(t) + 3 s(t) c(t) a, 9 x(t)^4 a^2 + 6 x(t)^3 c(t) a - 9 x(t)^2 a^2$$
$$+ x(t)^2 - 6 x(t) c(t) a - c(t)^2, s(t)^2 + c(t)^2 - 1]]$$

The constraints on the initial values

The complete set of constraints on the initial values

```
> iv_constraints :=
     Equations (result[1], order=0, notation=jet, solved);
```

$$iv\_constraints := \left[y = -\frac{-9 x^3 s a^2 - 3 x^2 s c a + x s c^2 + 9 x s a^2 - x s + 3 s c a}{c^3 - c}, x^4 = -\frac{1}{9} \frac{6 x^3 c a - 9 x^2 a^2 + x^2 - 6 x c a - c^2}{a^2}, s^2 = -c^2 + 1 \right] \quad \textbf{(1.1)}$$

Let us check if the initial values chosen above are consistent

```
> iv;
```

$$x(0) = 1, y(0) = 0, z(0) = 1 \quad \textbf{(1.2)}$$

Computing the initial values is easy since **iv_constraints is a triangular** polynomial system
Let us start from z(0) = 1.

```
> initial_values := s = sin(2.5), c = cos (2.5);
```

$$initial\_values := s = 0.5984721441, c = -0.8011436155 \quad \textbf{(1.3)}$$

The chosen initial values are not consistent: x(0) = 1 is not allowed

```
> x_values := [fsolve (subs (initial_values, a=0.7, b=2.5,
  iv_constraints[2]))];
```

$$x\_values := [-0.9787496359, 0.2938769208, 0.5894498759, 0.8584167589] \quad \textbf{(1.4)}$$

Let us compute some truly consistent initial values

```
> initial_values := initial_values, x = x_values[1];
```

$initial\_values := s = 0.5984721441, c = -0.8011436155, x = -0.9787496359$ **(1.5)**

```
> initial_values :=
     initial_values,
     subs (initial_values, a=0.7, b=2.5, iv_constraints[-3]);
```

$initial\_values := s = 0.5984721441, c = -0.8011436155, x = -0.9787496359,$ **(1.6)**

$y = -0.2050588960$

Let us check that the initial algebraic constraint is satisfied by the initial values

```
> subs (initial_values, x^2+y^2);
```

1.000000001 **(1.7)**

```
> consistent_iv :=
          x(0) = subs (initial_values, x),
          y(0) = subs (initial_values, y),
          z(0) = 1;
```

$consistent\_iv := x(0) = -0.9787496359, y(0) = -0.2050588960, z(0) = 1$ **(1.8)**

The DAE solver of MAPLE does not complain anymore

```
> sol := dsolve ({ op (num_book_syst), consistent_iv }, vars,
  numeric);
```

$sol :=$ **proc**(x_rkf45_dae) ... **end proc** **(1.9)**

```
> odeplot (sol, [t,x(t)], t=0..0.1);
```



The underlying explicit ODE

The missing ODE is part of the equations.
One can use it to allow the use of an explicit Runge–Kutta method

```
> missing_ODE :=
    diff (z(t),t) = NormalForm (diff (z(t),t), result[1]);
```

$$missing\_ODE := \frac{d}{dt} z(t) = \frac{1}{9\, c(t)^5\, a^2\, b - 9\, c(t)^3\, a^2\, b + c(t)\, b} \big(-243\, x(t)^3\, c(t)^4\, a^4 + 243\, x(t)^3\, c(t)^2\, a^6 + 108\, x(t)^3\, c(t)^2\, a^4 - 27\, x(t)^3\, a^4 - 9\, x(t)^3\, a^2 - 162\, x(t)^2\, c(t)^5\, a^3 + 81\, x(t)^2\, c(t)^3\, a^5 + 126\, x(t)^2\, c(t)^3\, a^3 - 27\, x(t)^2\, c(t)\, a^3 - 6\, x(t)^2\, c(t)\, a + 162\, x(t)\, c(t)^4\, a^4 + 9\, x(t)\, c(t)^4\, a^2 - 243\, x(t)\, c(t)^2\, a^6 - 27\, x(t)\, c(t)^2\, a^4 - 12\, x(t)\, c(t)^2\, a^2 + 27\, x(t)\, a^4 + 6\, x(t)\, a^2 - x(t) + 108\, c(t)^5\, a^3 - 81\, c(t)^3\, a^5 - 54\, c(t)^3\, a^3 - 6\, c(t)^3\, a + 9\, c(t)\, a^3 + 7\, c(t)\, a \big) \qquad (2.1)$$

The same ODE, with the initial notations

```
> missing_ODE :=
    subs (a=0.7, b=2.5, c(t)=cos(2.5*z(t)), s(t)=sin(2.5*z(t)),
  missing_ODE);
```

$$missing\_ODE := \frac{d}{dt} z(t) = (-58.3443\, x(t)^3 \cos(2.5\, z(t))^4 + 54.519507\, x(t)^3 \cos(2.5\, z(t))^2 - 10.8927\, x(t)^3 - 55.566\, x(t)^2 \cos(2.5\, z(t))^5 + 56.83167\, x(t)^2 \cos(2.5\, z(t))^3 - 13.461\, x(t)^2 \cos(2.5\, z(t)) + 43.3062\, x(t) \cos(2.5\, z(t))^4 - 40.951407\, x(t) \cos(2.5\, z(t))^2 + 8.4227\, x(t) + 37.044 \cos(2.5\, z(t))^5 - 36.33567 \cos(2.5\, z(t))^3 + 7.987 \cos(2.5\, z(t))) / (11.025 \cos(2.5\, z(t))^5 - 11.025 \cos(2.5\, z(t))^3 + 2.5 \cos(2.5\, z(t))) \quad \textbf{(2.2)}$$

The underlying ODE is obtained by removing the algebraic constraint and inserting the missing ODE

```
> explicit_syst := [ num_book_syst[1], num_book_syst[2],
  missing_ODE ];
```

$$explicit\_syst := \left[\frac{d}{dt} x(t) = 0.7\, y(t) + \sin(2.5\, z(t)), \frac{d}{dt} y(t) = 1.4\, x(t) + \cos(2.5\, z(t)), \frac{d}{dt} z(t) = (-58.3443\, x(t)^3 \cos(2.5\, z(t))^4 + 54.519507\, x(t)^3 \cos(2.5\, z(t))^2 - 10.8927\, x(t)^3 - 55.566\, x(t)^2 \cos(2.5\, z(t))^5 + 56.83167\, x(t)^2 \cos(2.5\, z(t))^3 - 13.461\, x(t)^2 \cos(2.5\, z(t)) + 43.3062\, x(t) \cos(2.5\, z(t))^4 - 40.951407\, x(t) \cos(2.5\, z(t))^2 + 8.4227\, x(t) + 37.044 \cos(2.5\, z(t))^5 - 36.33567 \cos(2.5\, z(t))^3 + 7.987 \cos(2.5\, z(t))) / (11.025 \cos(2.5\, z(t))^5 - 11.025 \cos(2.5\, z(t))^3 + 2.5 \cos(2.5\, z(t))) \right] \quad \textbf{(2.3)}$$

A basic Runge–Kutta scheme permits to integrate the **underlying ODE** system

```
> sol := dsolve ({ op (explicit_syst), consistent_iv }, vars,
  numeric);
```

$sol :=$ **proc**(x_rkf45) ... **end proc** **(2.4)**

```
> odeplot (sol, [t,x(t)], t = 0..0.1);
```


t
0
0.02
0.04
0.06
0.08
0.10
x
-0.92
-0.93
-0.94
-0.95
-0.96
-0.97